

Funktionen - erweitert

Excel 2019

Dagmar Serb

V.01

1	FUNKTIONEN – ERWEITERT	1
1.1	MATHEMATISCHE FUNKTIONEN.....	1
1.1.1	<i>DBSUMME</i>	1
1.2	STATISTISCHE FUNKTIONEN.....	2
1.2.1	<i>ZÄHLENWENN</i>	2
1.2.2	<i>ZÄHLENWENNNS</i>	3
1.3	TEXT-FUNKTIONEN	4
1.3.1	<i>LINKS</i>	4
1.3.2	<i>RECHTS</i>	4
1.3.3	<i>TEIL</i>	5
1.3.4	<i>ERSETZEN</i>	5
1.3.5	<i>WECHSELN</i>	6
1.3.6	<i>SUCHEN</i>	6
1.3.7	<i>FINDEN</i>	7
1.3.8	<i>VERKETTEN</i>	7
1.3.9	<i>TEXT</i>	8
1.3.10	<i>LÄNGE</i>	9
1.3.11	<i>Verschachteln Beispiel 1</i>	9
1.3.12	<i>Verschachteln Beispiel 2</i>	10
1.3.13	<i>Verschachteln Beispiel 3</i>	11

1 Funktionen – erweitert

Die angeführten Beispiele finden Sie in der Übungsdatei Funktionen-erweitert.xlsx

1.1 MATHEMATISCHE Funktionen

1.1.1 DBSUMME

Mit dieser Funktion kann man **Summen bilden**, die **bestimmten Suchkriterien** entsprechen.

Syntax	
DBSUMME(Datenbank;Datenbankfeld;Suchkriterien)	

Argumente	Beschreibung
Datenbank (erforderlich)	Der Zellbereich , aus dem die Liste oder Datenbank besteht. Die erste Zeile der Liste enthält Spaltenbeschriftungen .
Datenbankfeld (erforderlich)	Gibt an, welche Spalte in der Funktion verwendet wird . Sie können dafür die Spaltenbeschriftung in Anführungszeichen angeben (z.B. „Rabatt“ oder eine Zahl (ohne Anführungszeichen), die die Position der Spalte in der Liste darstellt : 1 für die erste Spalte, 2 für die zweite Spalte usw.
Suchkriterien (erforderlich)	Der Zellbereich , der die angegebenen Bedingungen enthält . Sie können jeden Bereich verwenden, der mindestens eine Spaltenbeschriftung und eine Zelle unter der Beschriftung zum Angeben der Bedingung enthält.

Beispiel: Alle Umsätze der Kunden aus Deutschland, mit denen ein Jahresumsatz von mehr als 500.000 € erzielt wurde und dabei die Anzahl der Aufträge bei mehr als 5 liegt, sollen in Zelle C10 aufsummiert werden:

	A	B	C	D
1	Kunde	Land	Jahresumsatz	Aufträge
2	Kaiser	Deutschland	€ 750.567,00	6
3	Beckmann	Schweiz	€ 289.675,00	3
4	Müller	Deutschland	€ 118.514,00	2
5	Rickmers	Deutschland	€ 861.536,00	5
6				
7	Analyse			
8	Kunde	Land	Jahresumsatz	Aufträge
9		Deutschland	>500000	>5
10			=DBSUMME(A1:D5;C8;A8:D9)	

Die Formel:

- > **Datenbank:** Zellbereich A1 bis D5 (Datenbank inkl. Spaltenüberschriften!)
- > **Datenbankfeld:** Zelle C8 (Jahresumsatz, welcher summiert werden soll. Kann in der Formel auch als Text eingetragen werden: "Jahresumsatz").



- › **Suchkriterien: Zellbereich A8 bis D9** (! Überschriften im Suchkriterien-Bereich müssen ident sein mit denen im Datenbank-Bereich. Es müssen nicht alle Felder aus dem Datenbank-Bereich angegeben werden!)

1.2 STATISTISCHE Funktionen

1.2.1 ZÄHLENWENN

Diese Funktion **zählt alle Werte**, die **das Suchkriterium erfüllen**.

Syntax
=ZÄHLENWENN(Bereich; Suchkriterium)

Argumente	Beschreibung
Bereich (erforderlich)	Bereich , in dem die Zellen gezählt werden. Leere Zellen und Textwerte werden ignoriert.
Suchkriterium (erforderlich)	Suchkriterium, welches im Bereich gezählt wird . Es kann eine Zahl, ein Ausdruck, ein Zellbezug oder eine Textzeichenfolge sein. <i>Z.B.: Zahl 32, Vergleich ">32", Bezug B2 oder "Text"</i> .

Ein paar Beispiele:

	A	B	C	D	E	F	G
1	Sorte	Kisten					
2	Apfel	12					
3	Birne	16					
4	Kirsche	8					
5	Apfel	36					
6	Birne	13					
7	Apfel	14					
8							
a)	Anzahl der Zellen mit dem Wert Apfel:			3	=ZÄHLENWENN(A2:A7;"Apfel")		
b)	Anzahl der Zellen mit einem Wert größer als 15:			2	=ZÄHLENWENN(B2:B7;">15")		
c)	Anzahl der Zellen mit dem Wert Apfel und Birne:			5	=ZÄHLENWENN(A2:A7;A2)+ZÄHLENWENN(A2:A7;A3)		
d)	Anzahl der Zellen mit beliebigem Text:			6	=ZÄHLENWENN(A2:A7;"*")		
e)	Anzahl der Zellen mit 5 Zeichen und "el" als letzten beiden Zeichen:			3	=ZÄHLENWENN(A2:A7;"???el")		

- a) **Bereich: A2 bis A7 (Sorten), Suchkriterium: "Apfel"**. Das Ergebnis lautet „3“.
- b) **Bereich: B2 bis B7 (Anzahl Kisten), Suchkriterium: ">15"**. Das Ergebnis lautet „2“.
- c) **Bereich: A2 bis A7 (Sorten), Suchkriterium: "Apfel" und nach gleichem Schema wird zweite Funktion für „Birne“ mit Pluszeichen (+) angehängt**. Das Ergebnis lautet „5“.
- d) **Bereich: A2 bis A7 (Sorten), Suchkriterium: "*" (, "*" wird als Platzhalterzeichen für eine beliebige Zeichenfolge verwendet)**. Das Ergebnis lautet „6“.
- e) **Bereich: A2 bis A7 (Sorten), Suchkriterium: "???el" (, "?" wird als Platzhalterzeichen für ein einzelnes Zeichen verwendet)**. Das Ergebnis lautet „3“.

1.2.2 ZÄHLENWENNS

Diese Funktion **zählt alle Werte**, die **die Suchkriterien erfüllen**.

Syntax

=ZÄHLENWENNS(Kriterienbereich1;Kriterien1;[Kriterienbereich2; Kriterien2]...)

Argumente	Beschreibung
Kriterienbereich 1 (erforderlich)	Erster Bereich , in dem die zugehörigen Kriterien ausgewertet werden.
Kriterien 1 (erforderlich)	Kriterium, welches im entsprechenden Bereich gezählt wird . Es kann eine Zahl, ein Ausdruck, ein Zellbezug oder eine Textzeichenfolge sein. <i>Z.B.: Zahl 32, Vergleich ">32", Bezug B2 oder "Text"</i> .
Kriterienbereich2; Kriterien2; ... (optional)	Zusätzliche Bereiche und deren zugehörige Kriterien. Es sind bis zu 127 Bereich/Kriterien-Paare zulässig.

In diesem Beispiel wird gezählt, wie viele Verkäufer sowohl Bonus 1 als auch Bonus 2 erhalten:

	A	B	C	D	E
1	Verkäufer	Bonus 1	Bonus 2	Bonus 3	
2	Maier	ja	nein	nein	
3	Müller	ja	ja	nein	
4	Hofer	ja	ja	ja	
5	Bauer	nein	ja	ja	
6					
7	Anzahl Verkäufer, die sowohl Bonus 1 als auch Bonus 2 erhalten:			2	

Die Formel:

- > **Kriterienbereich 1:**
B2 bis B5 (Werte Bonus1)
 - > **Kriterien 1:** „ja“
 - > **Kriterienbereich 2:**
C2 bis C5 (Werte Bonus2)
 - > **Kriterien 2:** „ja“
- Das Ergebnis lautet „2“

Dieses Beispiel ermittelt, in wie vielen Regionen im Jänner mehr als 25.000 Euro und im März mehr als 23.000 Euro erwirtschaftet wurden:

	A	B	C	D	E
1	Verkäufer	Jänner	Februar	März	
2	Region 1	€ 26.000,00	€ 25.000,00	€ 27.000,00	
3	Region 2	€ 29.000,00	€ 25.000,00	€ 21.000,00	
4	Region 3	€ 23.000,00	€ 30.000,00	€ 24.000,00	
5					
6	Anzahl Regionen, in denen im Jänner mehr als 25.000 Euro und im März mehr als 23.000 Euro erwirtschaftet wurden:			1	

Die Formel:

- > **Kriterienbereich 1:**
B2 bis B4
(Werte Jänner)
 - > **Kriterien 1:**
„>25000“
 - > **Kriterienbereich 2:**
D2 bis D4
(Werte März)
 - > **Kriterien 2:**
„>23000“
- Das Ergebnis lautet „1“



1.3 TEXT-Funktionen

Textfunktionen können **separat** verwendet werden, kommen in der Praxis aber häufig **in Kombination miteinander** vor. Sogenannte **Verschachtelungen** braucht man zum Beispiel dazu, um aus Listen **Daten herauszuziehen** (*PLZ und Ort trennen*). Zunächst aber die Funktionen einzeln erklärt:

1.3.1 LINKS

Diese Funktion **gibt** je nach Angabe **das oder die erste(n) Zeichen** einer Textzeichenfolge zurück.

Syntax
=LINKS(Text;[Anzahl_Zeichen])

Argumente	Beschreibung
Text (erforderlich)	Text , der zu extrahieren ist.
Anzahl_Zeichen (optional)	Gibt die Anzahl der Zeichen an, die zu extrahieren sind. <ul style="list-style-type: none">Anzahl_Zeichen muss größer gleich 0 sein.Ist Anzahl_Zeichen größer als „Text“, wird der gesamte Text zurückgegeben.Fehlt Anzahl_Zeichen, wird es als 1 angenommen.

Beispiel:

	A	B	C	D
1	Daten	Ergebnis	Formel	Beschreibung
2	8700 Leoben	8700	=LINKS(A2;4)	Extrahiert die ersten vier Zeichen der Zeichenfolge in A2.
3	8600 Bruck/Mur	8600	=LINKS(A3;4)	Extrahiert die ersten vier Zeichen der Zeichenfolge in A3.
4	8650 Kindberg	8650	=LINKS(A4;4)	Extrahiert die ersten vier Zeichen der Zeichenfolge in A4.

1.3.2 RECHTS

Diese Funktion **gibt** je nach Angabe **das oder die letzte(n) Zeichen** einer Textzeichenfolge zurück.

Syntax
=RECHTS(Text;[Anzahl_Zeichen])

Argumente	Beschreibung
Text (erforderlich)	Text , der zu extrahieren ist.
Anzahl_Zeichen (optional)	Gibt die Anzahl der Zeichen an, die zu extrahieren sind. <ul style="list-style-type: none">Anzahl_Zeichen muss größer gleich 0 sein.Ist Anzahl_Zeichen größer als „Text“, wird der gesamte Text zurückgegeben.Fehlt Anzahl_Zeichen, wird es als 1 angenommen.

Beispiel:

	A	B	C	D
1	Daten	Ergebnis	Formel	Beschreibung
2	Artikel001	001	=RECHTS(A2;3)	Extrahiert die letzten drei Zeichen der Zeichenfolge in A2 .
3	Artikel002	002	=RECHTS(A3;3)	Extrahiert die letzten drei Zeichen der Zeichenfolge in A3 .
4	Artikel003	003	=RECHTS(A4;3)	Extrahiert die letzten drei Zeichen der Zeichenfolge in A4 .

1.3.3 TEIL

Diese Funktion **extrahiert** aus einem Text je nach Angabe **eines oder mehrere Zeichen**.

Syntax

=TEIL(Text;Erstes_Zeichen;Anzahl_Zeichen)

Argumente	Beschreibung
Text (erforderlich)	Text , der zu extrahieren ist.
Erstes_Zeichen (erforderlich)	Gibt an, ab welchem Zeichen der Inhalt extrahiert wird.
Anzahl_Zeichen (erforderlich)	Gibt die Anzahl der zu extrahierenden Zeichen an.

Beispiel:

	A	B	C	D
1	Daten	Ergebnis	Formel	Beschreibung
2	8700 Leoben	Leoben	=TEIL(A2;6;50)	Extrahiert aus der Zeichenfolge in A2 ab dem 6. Zeichen 50 Zeichen.
3	8600 Bruck/Mur	Bruck/Mur	=TEIL(A3;6;50)	Extrahiert aus der Zeichenfolge in A3 ab dem 6. Zeichen 50 Zeichen.
4	8650 Kindberg	Kindberg	=TEIL(A4;6;50)	Extrahiert aus der Zeichenfolge in A4 ab dem 6. Zeichen 50 Zeichen.

1.3.4 ERSETZEN

Diese Funktion **ersetzt** innerhalb eines Texts **eine an einer bestimmten Position beginnende Zeichenfolge**.

Syntax

=ERSETZEN(Alter_Text;Erstes_Zeichen;Anzahl_Zeichen;Neuer_Text)

Argumente	Beschreibung
Alter_Text (erforderlich)	Text , bei dem eine Anzahl von Zeichen ersetzt wird.
Erstes_Zeichen (erforderlich)	Position des Zeichens in "Alter_Text", an der mit dem Ersetzen begonnen wird.
Anzahl_Zeichen (erforderlich)	Anzahl der Zeichen in "Alter_Text", die durch "Neuer_Text" ersetzt werden.
Neuer_Text (erforderlich)	Text , der die Zeichen in "Alter_Text" ersetzt .



Beispiele:

	A	B	C	D
1	Daten	Ergebnis	Formel	Beschreibung
2	2019	2020	=ERSETZEN(A1;3;2;"20")	Ersetzt die beiden letzten Ziffern von 2019 durch 20
3	m.muster@excel.at	m.muster(at)excel.at	=ERSETZEN(A2;9;1;"(at)")	Ersetzt das neunte Zeichen von "m.muster@excel.at" durch "(at)".

1.3.5 WECHSELN

Diese Funktion **ersetzt** innerhalb eines Textes **eine bestimmte Zeichenfolge**.

Syntax
WECHSELN(Text;Alter_Text;Neuer_Text;[ntes_Auftreten])

Argumente	Beschreibung
Text (erforderlich)	Text (in Hochkommas ("")!) oder Zellbezug , in dem Zeichen ausgetauscht werden.
Alter_Text (erforderlich)	Text , der ersetzt wird (in Hochkommas ("")!).
Neuer_Text (erforderlich)	Text , der „Alter_Text“ ersetzt (in Hochkommas ("")!).
ntes_Auftreten (optional)	Gibt an, an welchen Stellen „Alter_Text“ durch „Neuer_Text“ ersetzt wird. Wird „ntes_Auftreten“ nicht angegeben , werden alle Vorkommen von „Alter_Text“ ersetzt .

Beispiele:

	A	B	C	D
1	Daten	Ergebnis	Formel	Beschreibung
2	Umsatzdaten	Verkaufsdaten	=WECHSELN(A2;"Umsatz";"Verkaufs")	Wechselt "Umsatz" in "Verkaufs".
3	Quartal 2, 2019	Quartal 3, 2019	=WECHSELN(A3;"2";"3";1)	Wechselt die erste Instanz von "2" in "3".
4	Quartal 1, 2011	Quartal 1, 2019	=WECHSELN(A4;"1";"9";3)	Wechselt die dritte Instanz von "1" in "9".

1.3.6 SUCHEN

Diese Funktion **sucht nach dem ersten Vorkommen einer Zeichenfolge innerhalb einer anderen Zeichenfolge** und gibt die **Position der Fundstelle als Nummer** zurück. **Groß- und Kleinschreibung** ist **irrelevant**, die Platzhalterzeichen „?“ und „*“ werden **unterstützt**.

Syntax
=SUCHEN(Suchtext;Text;[Erstes_Zeichen])

Argumente	Beschreibung
Suchtext (erforderlich)	Text , der zu finden ist.
Text (erforderlich)	Text , in dem „Suchtext“ gesucht wird.
Erstes_Zeichen (optional)	Gibt an, bei welchem Zeichen die Suche begonnen wird. Das erste Zeichen hat die Nummer 1. Fehlt das Argument „Erstes_Zeichen“, wird es als 1 angenommen.

Beispiele:

	A	B	C	D
1	Daten	Ergebnis	Formel	Beschreibung
2	Frank Mc Court	8	=SUCHEN("C";A2)	Ermittelt die Position des ersten "C" in der Zeichenfolge. Ergebnis: "8"
3	Fahrradfahren	1	=SUCHEN("fahr";A3)	Ermittelt die Position des ersten "fahr" in der Zeichenfolge. Ergebnis: "1"
4	Fahrradfahren	4	=SUCHEN("r*n";A4)	Ermittelt die Position der ersten Zeichenfolge, die mit "r" beginnt und mit "n" endet. Dazwischen dürfen sich mehrere beliebige Zeichen befinden. Ergebnis: "4"
5	Fahrradfahren	11	=SUCHEN("r?n";A5)	Ermittelt die Position der ersten Zeichenfolge, die mit "r" beginnt und mit "n" endet. Dazwischen darf sich ein beliebiges Zeichen befinden. Ergebnis: "11"

Im Unterschied zu *SUCHEN* unterscheidet *FINDEN* zwischen Groß- und Kleinschreibung und unterstützt keine Platzhalterzeichen:

1.3.7 FINDEN

Diese Funktion **sucht nach dem ersten Vorkommen einer Zeichenfolge innerhalb einer anderen Zeichenfolge** und gibt dabei die **Position der Fundstelle als Nummer** zurück. *FINDEN* unterscheidet zwischen Groß- und Kleinschreibung, Platzhalter sind nicht erlaubt.

Syntax

=FINDEN(Suchtext;Text;[Erstes_Zeichen])

Argumente	Beschreibung
Suchtext (erforderlich)	Text , der zu finden ist.
Text (erforderlich)	Text , in dem „Suchtext“ gesucht wird.
Erstes_Zeichen (optional)	Gibt an, bei welchem Zeichen die Suche begonnen wird. Das erste Zeichen hat die Nummer 1. Fehlt das Argument „Erstes_Zeichen“, wird es als 1 angenommen.

Beispiele:

	A	B	C	D
1	Daten	Ergebnis	Formel	Beschreibung
2	Frank Mc Court	10	=FINDEN("C";A2)	Ermittelt die Position des ersten "C" in der Zeichenfolge. Ergebnis: "10"
3	Fahrradfahren	8	=FINDEN("fahr";A3)	Ermittelt die Position des ersten "fahr" in der Zeichenfolge. Ergebnis: "8"

Im Unterschied zu *FINDEN* unterscheidet *SUCHEN* nicht zwischen Groß- und Kleinschreibung und die Verwendung von Platzhalterzeichen ist erlaubt!

1.3.8 VERKETTEN

Diese Funktion **verbindet** zwei oder mehrere **Zeichenfolgen zu einer Zeichenfolge**.

Syntax

= VERKETTEN(Text1;[Text2]; ...)



Argumente	Beschreibung
Text1 (erforderlich)	Die erste zu verknüpfende Zeichenfolge .
Text2, ... (optional)	Weitere , bis zu 255 verknüpfende Zeichenfolgen .

Beispiele:

	A	B	C	D	E
1	Daten	Daten	Daten	Ergebnis	Formel
2	Frank	Mc Court		Frank Mc Court	=VERKETTEN(A2;" ";B2)

Verkettet wird A2 (Frank) mit B2 (Mc Court). Das Leerzeichen zwischen Vor- und Nachnamen wird in der Formel zwischen Hochkommas (" ") eingegeben.

	A	B	C	D	E
1	Daten	Daten	Daten	Ergebnis	Formel
3	Frank	Mc Court	Buchautor	Frank Mc Court ist ein fantastischer Buchautor	=VERKETTEN(A3;" ";B3;" "; "ist ein fantastischer";" ";C3)

Verkettet wird A2 (Frank) mit B2 (Mc Court). Das Leerzeichen zwischen Vor- und Nachnamen wird in der Formel zwischen Hochkommas (" ") eingegeben. Es folgen ein Leerzeichen und der Text „ist ein fantastischer“ (in Hochkommas!). Nach einem weiteren Leerzeichen wird „Buchautor“ als Zellbezug C2 eingefügt.

😊 **Tip:** Mit dem **kaufmännischen Und-Zeichen (&)** als Berechnungsoperator können Sie Textelemente verknüpfen, **ohne eine Funktion** verwenden zu müssen!

	A	B	C	D	E
1	Daten	Daten	Ergebnis	Formel	Beschreibung
2	Frank	Mc Court	Frank Mc Court	=A2&" "&B2	A2 (Frank) und B2 (Mc Court) werden mit "&" verknüpft. Das Leerzeichen zwischen Vor- und Nachname wird in Hochkommas (" ") eingegeben.
3	H	M	H&M	=A3&" "&B3	A3 (H) und B3 (M) werden mit "&" verknüpft. Das "&" zwischen "H" und "M" wird in Hochkommas (" ") eingegeben.

1.3.9 TEXT

Diese Funktion **wandelt einen numerischen Wert in Text um**, wobei gleichzeitig das **Anzeigeformat** festgelegt wird.

Syntax
TEXT(Wert;Textformat)

Argumente	Beschreibung
Wert (erforderlich)	Numerischer Wert , Formel , die einen numerischen Wert liefert, oder Bezug auf eine Zelle, die einen numerischen Wert enthält.
Textformat (erforderlich)	Numerisches Format als eine Zeichenfolge in Anführungszeichen (z. B. "t.M.jjjj" oder "#.#0,00").

Beispiele:

	A	B	C	D
1	Daten	Ergebnis	Formel	Beschreibung
2	5,4	5,400	=TEXT(A2;"#,000")	Gibt den Wert in Zelle A2 mit drei Nachkommastellen aus.
3	29,895	29,90 €	=TEXT(A3;"#.###,00 €")	Gibt den Wert in Zelle A3 mit Tausendertrennpunkt, 2 Nachkommastellen und dem Eurozeichen aus.
4	5,25	5 1/4	=TEXT(A4;"# ??/???")	Gibt die Nachkommastelle des Werts in A4 als Bruch aus.
5	09.Dez	09.12.2015	=TEXT(A5;"tt.MM.jjjj")	Zeigt das Datum in A5 wie folgt an: Tag mit vorangestellter 0, Monat falls zutreffend mit vorangestellter 0, Jahreszahl vierstellig.

1.3.10 LÄNGE

Diese Funktion gibt die **Anzahl der Zeichen** einer Zeichenfolge zurück.

Syntax
LÄNGE(Text)

Argumente	Beschreibung
Text (erforderlich)	Text , dessen Länge ermittelt werden soll. <i>Leerzeichen zählen als Zeichen.</i>

Beispiel:

	A	B	C	D
1	Daten	Ergebnis	Formel	Beschreibung
2	Meine Überschrift	17	=LÄNGE(A2)	Gibt die Anzahl der Zeichen in Zelle A2 zurück.

1.3.11 Verschachteln Beispiel 1

Im Beispiel der **Funktion LINKS** wurden Postleitzahlen von Orten getrennt. Ist die **Anzahl der zu extrahierenden Zeichen variabel** (österreichische PLZ sind 4stellig, deutsche 5stellig!), wird **zusätzlich** die **Funktion FINDEN** benötigt, um die **Stelle für die Extraktion zu ermitteln**.

B2	
=LINKS(A2;FINDEN("-",A2)-1)	

	A	B	C
1	Daten	Ergebnis	
2	8700-Leoben	8700	
3	20095-Hamburg	20095	
4	8600-Bruck/Mur	8600	



Die Syntaxen der beiden benötigten Funktionen:

=LINKS(Text;Anzahl_Zeichen)
=FINDEN(Suchtext;Text;[Erstes_Zeichen])

Funktion	Erklärung
=LINKS(Einleiten der Funktion LINKS
=LINKS(A2;	TEXT: Text in A2, der zu extrahieren ist
=LINKS(A2; FINDEN(ANZAHL_ZEICHEN: Anzahl der zu extrahierenden Zeichen. Dazu wird die Funktion FINDEN benötigt: Sie ermittelt die Position des Bindestrichs, die als eindeutige Trennstelle zwischen PLZ und Ort gesehen werden kann.
=LINKS(A2; FINDEN("-";	SUCHTEXT: Text "-", der zu finden ist
= LINKS (A2; FINDEN("-";A2); =	TEXT: Text, in dem gesucht wird – A2
LINKS (A2; FINDEN("-";A2)-1)	-1, damit der Bindestrich nicht mit extrahiert wird

1.3.12 Verschachteln Beispiel 2

Aus „Reingewinn 2018“ soll in Zelle B2 „Reinerlös 2018“ werden.

Mit der Funktion ERSETZEN wird "gewinn" durch "erlös" ersetzt, indem zuerst mit der Funktion SUCHEN nach der Position von "gewinn" gesucht und dann durch die Zeichenfolge "erlös" ersetzt wird.

	A	B	C
1	Daten	Ergebnis	Beschreibung
2	Reingewinn 2018	Reinerlös 2018	Ersetzt "Reingewinn" durch "Reinerlös"
3	gewinn		

Die Syntaxen der beiden benötigten Funktionen:

=ERSETZEN(Alter_Text;Erstes_Zeichen;Anzahl_Zeichen;Neuer_Text)
=SUCHEN(Suchtext;Text)

Funktion	Erklärung
=ERSETZEN(=ERSETZEN(A2; =ERSETZEN(A2; SUCHEN(=ERSETZEN(A2; SUCHEN(A3; =ERSETZEN(A2; SUCHEN(A3;A2); =ERSETZEN(A2; SUCHEN(A2;A3);6; =ERSETZEN(A2; SUCHEN(A2;A3);6; "erlös")	<p>Einleiten der Funktion ERSETZEN</p> <p>ALTER_TEXT: Text, bei dem ersetzt wird – A2</p> <p>ERSTES_ZEICHEN: Position, wo mit dem Ersetzen begonnen wird. Dazu wird die Funktion SUCHEN benötigt: sie sucht nach „gewinn“ und gibt dann die Position als Nummer zurück.</p> <p>SUCHTEXT: Text, der gesucht wird – A3</p> <p>TEXT: Text, in dem gesucht wird – A2</p> <p>ANZAHL_ZEICHEN: Anzahl der zu ersetzenden Zeichen</p> <p>NEUER_TEXT: der ersetzende Text</p>

☺ **Tipp:** Das Argument **Anzahl_Zeichen** können Sie auch mit der Funktion **LÄNGE** ermitteln:
 =ERSETZEN(A2;SUCHEN(A3;A2);LÄNGE(A3);"erlös")

1.3.13 Verschachteln Beispiel 3

Im nächsten Beispiel wird der errechnete **Artikelpreis** mithilfe der Funktionen **TEXT** und der Funktion **VERKETTEN** als **ausformulierter Satz wiedergegeben**. Das **Anzeigeformat** wird über die Funktion **TEXT** definiert.

	A	B	C	D	F	G	H
1	Preis Netto	MwSt.	Preis Brutto	Datum			
2	26 €	5 €	31 €	42347			
3							
4	Verketteten						
5	Der Artikel kostet 31,20 € inkl. MwSt.						
6	Methode "Kaufmännisches-Und (&)"						
7	Der Artikel kostet 31,20 € inkl. MwSt.						
8	Datum: 2015-12-09						

Die Syntaxen der beiden benötigten Funktionen:

=VERKETTEN(Text1; [Text2]; ...)

=TEXT(Wert;Textformat)



Funktion

=VERKETTEN(

=VERKETTEN("Der Artikel kostet";

=VERKETTEN("Der Artikel kostet";" ", //Leerzeichen

=VERKETTEN("Der Artikel kostet";" ";TEXT(C2;"#.##0,00 €"); //Verwendung der Funktion TEXT

=VERKETTEN("Der Artikel kostet";" ";TEXT(C2;"#.##0,00 €");" "; //Leerzeichen

=VERKETTEN("Der Artikel kostet";" ";TEXT(C2;"#.##0,00 €");" ";"inkl. MwSt.")

Sie können auch **auf die Funktion VERKETTEN verzichten** und mit dem **kaufmännischen Und (&)** arbeiten:

	A	B	C	D	E
1	Preis Netto	MwSt.	Preis Brutto	Datum	
2	26 €	5 €	31 €	42347	
3					
4	Verketteten			Formel	
5	Der Artikel kostet 31,20 € inkl. MwSt.			=VERKETTEN("Der Artikel kostet";" ";TEXT(C2;"#.##0,00 €");" ";"inkl. MwSt.")	
6	Methode "Kaufmännisches-Und (&)"				
7	Der Artikel kostet 31,20 € inkl. MwSt.			="Der Artikel kostet "& TEXT(C2;"#.##0,00 €") &" inkl. MwSt."	
8	Datum: 2015-12-09			="Datum: " & TEXT(D2;"jjjj-MM-tt")	

Kaufmännisches Und (&)

="Der Artikel kostet "& TEXT(C2;"#.##0,00 €") &" inkl. MwSt."

="Datum: " & TEXT(D2;"jjjj-MM-tt") ergibt: Datum: 2015-12-09

